

NAG Fortran Library Routine Document

F07BEF (SGBTRS/DGBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07BEF (SGBTRS/DGBTRS) solves a real band system of linear equations with multiple right-hand sides, $AX = B$ or $A^T X = B$, where A has been factorized by F07BDF (SGBTRF/DGBTRF).

2 Specification

```

SUBROUTINE F07BEF(TRANS, N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB, INFO)
ENTRY      sgbtrs (TRANS, N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB, INFO)
INTEGER    N, KL, KU, NRHS, LDAB, IPIV(*), LDB, INFO
real      AB(LDAB,*), B(LDB,*)
CHARACTER*1 TRANS

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a real band system of linear equations $AX = B$ or $A^T X = B$, this routine must be preceded by a call to F07BDF (SGBTRF/DGBTRF) which computes the LU factorization of A as $A = PLU$. The solution is computed by forward and backward substitution.

If TRANS = 'N', the solution is computed by solving $PLY = B$ and then $UX = Y$.

If TRANS = 'T' or 'C', the solution is computed by solving $U^T Y = B$ and then $L^T P^T X = Y$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: TRANS – CHARACTER*1 *Input*
On entry: indicates the form of the equations as follows:
 if TRANS = 'N', $AX = B$ is solved for X ;
 if TRANS = 'T' or 'C', $A^T X = B$ is solved for X .
Constraint: TRANS = 'N', 'T' or 'C'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: KL – INTEGER *Input*
On entry: k_l , the number of sub-diagonals within the band of A .
Constraint: $KL \geq 0$.

- 4: KU – INTEGER *Input*
On entry: k_u , the number of super-diagonals within the band of A .
Constraint: $KU \geq 0$.
- 5: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides.
Constraint: $NRHS \geq 0$.
- 6: AB(LDAB,*) – *real* array *Input*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the LU factorization of A , as returned by F07BDF (SGBTRF/DGBTRF).
- 7: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F07BEF (SGBTRS/DGBTRS) is called.
Constraint: $LDAB \geq 2 \times KL + KU + 1$.
- 8: IPIV(*) – INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
On entry: the pivot indices, as returned by F07BDF (SGBTRF/DGBTRF).
- 9: B(LDB,*) – *real* array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the n by r right-hand side matrix B .
On exit: the n by r solution matrix X .
- 10: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07BEF (SGBTRS/DGBTRS) is called.
Constraint: $LDB \geq \max(1, N)$.
- 11: INFO – INTEGER *Output*
On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If $INFO = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(k)\epsilon P|L||U|,$$

$c(k)$ is a modest linear function of $k = k_l + k_u + 1$, and ϵ is the *machine precision*. This assumes $k \ll n$.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(k) \text{cond}(A, x)\epsilon$$

where $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty} \leq \text{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$. Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$, and $\text{cond}(A^T)$ can be much larger (or smaller) than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07BHF (SGBRFS/DGBRFS), and an estimate for $\kappa_{\infty}(A)$ can be obtained by calling F07BGF (SGBCON/DGBCON) with NORM = 'I'.

8 Further Comments

The total number of floating-point operations is approximately $2n(2k_l + k_u)r$, assuming $n \gg k_l$ and $n \gg k_u$.

This routine may be followed by a call to F07BHF (SGBRFS/DGBRFS) to refine the solution and return an error estimate.

The complex analogue of this routine is F07BSF (CGBTRS/ZGBTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 & 0.00 \\ -6.98 & 2.46 & -2.73 & -2.13 \\ 0.00 & 2.56 & 2.46 & 4.07 \\ 0.00 & 0.00 & -4.78 & -3.82 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 4.42 & -36.01 \\ 27.13 & -31.67 \\ -6.14 & -1.16 \\ 10.50 & -25.82 \end{pmatrix}.$$

Here A is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BDF (SGBTRF/DGBTRF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07BEF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
INTEGER          NMAX, KLMAX, KUMAX, LDAB, NRHMAX, LDB
PARAMETER       (NMAX=8, KLMAX=8, KUMAX=8, LDAB=2*KLMAX+KUMAX+1,
+              NRHMAX=NMAX, LDB=NMAX)
CHARACTER       TRANS
PARAMETER       (TRANS='N')
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, K, KL, KU, N, NRHS
*      .. Local Arrays ..
real           AB(LDAB, NMAX), B(LDB, NRHMAX)
INTEGER          IPIV(NMAX)
*      .. External Subroutines ..
EXTERNAL        sgbtrf, sgbtrs, X04CAF
*      .. Intrinsic Functions ..
INTRINSIC       MAX, MIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07BEF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, NRHS, KL, KU
IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX .AND. KL.LE.KLMAX .AND. KU.LE.
+   KUMAX) THEN
*
*      Read A and B from data file
```

```

*
      K = KL + KU + 1
      READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Factorize A
*
      CALL sgbtrf(N,N,KL,KU,AB,LDAB,IPIV,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*          Compute solution
*
      CALL sgbtrs(TRANS,N,KL,KU,NRHS,AB,LDAB,IPIV,B,LDB,INFO)
*
*          Print solution
*
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
*
      ELSE
      WRITE (NOUT,*) 'The factor U is singular'
      END IF
      END IF
      STOP
*
      END

```

9.2 Program Data

F07BEF Example Program Data

```

  4  2  1  2           :Values of N, NRHS, KL and KU
-0.23  2.54 -3.66
-6.98  2.46 -2.73 -2.13
          2.56  2.46  4.07
          -4.78 -3.82       :End of matrix A
  4.42 -36.01
 27.13 -31.67
 -6.14 -1.16
 10.50 -25.82           :End of matrix B

```

9.3 Program Results

F07BEF Example Program Results

```

Solution(s)
          1          2
 1    -2.0000    1.0000
 2     3.0000   -4.0000
 3     1.0000    7.0000
 4    -4.0000   -2.0000

```
